Fake News and Asset Price Dynamics

Sarah Mignot, Paolo Pellizzari and Frank Westerhoff

```
(* The following code replicates the basic model dynamics *)

Clear["Global`*"];
(* model equations *)
PF[t_] := PF[t] = (DD - λ * Var[t] * QQ)/r;
P[t_] := P[t] = (X[t] + DD - λ * Var[t] * QQ)/(1 + r);
X[t_] := X[t] = WC[t] * XC[t] + WF[t] * XF[t];
XC[t_] := XC[t] = P[t - 1] + trend * (P[t - 1] - P[t - 2]);
XF[t_] := XF[t] = P[t - 1] + fund * (PF[t - 1] - P[t - 1]);
WC[t_] := WC[t] = Exp[β * FC[t]]/(Exp[β * FC[t]] + Exp[β * FF[t]]);
WF[t_] := WF[t] = 1 - WC[t];
FC[t_] :=
  FC[t] = (P[t - 1] + DD - (1 + r) * P[t - 2]) * DC[t - 2] - λ/2 * Var[t - 2] * DC[t - 2]^2;
FF[t_] := FF[t] = (P[t - 1] + DD - (1 + r) * P[t - 2]) * DF[t - 2] -
    λ/2 * Var[t - 2] * DF[t - 2]^2 + α * (P[t - 1] - PF[t - 1])^2 - COST[t - 1];
DC[t_] := DC[t] = (XC[t] + DD - (1 + r) * P[t])/(λ * Var[t]);
DF[t_] := DF[t] = (XF[t] + DD - (1 + r) * P[t])/(λ * Var[t]);
COST[t_] := COST[t] = c1 * (VD[t]) + c2 * VF[t] + c3 * VP[t];
Var[t_] := Var[t] = VD[t] + VF[t] + VP[t];
VD[t_] := VD[t] = σD;
VF[t_] := VF[t] = σF;
VP[t_] := VP[t] = mv * VP[t - 1] + (1 - mv) * (P[t - 1] - MU[t - 1])^2;
MU[t_] := MU[t] = μ * MU[t - 1] + (1 - μ) * P[t - 1];

(* parameter setting *)
r = 0.1; DD = 10; QQ = 1; σD = 2; σF = 5; λ = 0.1; mv = μ = 0.9;
trend = 1.2; fund = 0.8; β = 1; c1 = 0.5; c2 = 0.5; c3 = 0.5;
α = 1.0;

(* definitions and initial values *)
FSS = (DD - λ * (σD + σF) * QQ)/r;
P[1] = FSS + 0.05;
P[0] = P[-1] = P[-2] = FSS; MU[1] = MU[0] = MU[-1] = MU[-2] = FSS;
VP[1] = VP[0] = VP[-1] = VP[-2] = 0;
```

```
(* computations *)
Table[P[t], {t, 1, 5000}];
meanP = Mean[Table[P[t], {t, 4100, 5200}]];
ListLinePlot[{Table[P[t], {t, 5000, 5200}], Table[PF[t], {t, 5000, 5200}],
  Table[meanP, {t, 5000, 5200}], Table[FSS, {t, 5000, 5200}]},
 PlotRange → {{-1, 203}, {85.5, 98.5}}, Frame → True,
 PlotStyle → {Purple, Gray, Directive[Dashed, Orange],
   Directive[Dashed, Green]}, FrameLabel → {"time", "price"}, AspectRatio → 1]
```

Out[•]=